
 The new MTIME method enables simulations when data sets size 

otherwise would be a restricting factor. 

 

 Efficient TTE trial simulations were implemented in NM without losing 

precision in the event time simulations. 

 

 The RTTE MTIME implementation improves accuracy compared to 

the grid method. 

With the developed method, using the original data set, the NM code simulates 

event times and based on these a table output with the resulting dependent 

variable at the event time is generated similar to the output obtained with 

dense grid data set simulation. The method was implemented for 4 parametric 

TTE distributions/survival functions [4] presented in Table 1.  

 

 

 

 

 

 

 

 

 

 

 

The distributions in Table I were used to simulate time to first event for 1000 

subjects with I) analytic solutions, II) a dense grid using a data set with dummy 

observations daily up to day 100 for each subject with hazard functions 

implemented in $DES and III) a new approach using a data set with two 

observations per individual (at day 0 and censoring time/end of study at day 

100), with hazard functions implemented in $DES but employment of MTIME. 

Different MTIME (DT=0.1, 0.5, 1, 2, 10 days) were used to force the ODE 

solver to increase the precision in $DES. 

Simulations of repeated time-to-event (RTTE) were also investigated and 

compared the dense grid method using an exponential distribution and a data 

set with 120 subjects (grid resolution of 0.5 days and study censoring time at 

day 288) and MTIME with DT=0.001 days. Pseudo code of the 

implementations are available in Figure 2. 

$PK 

 IF (NEWIND.NE.2) THEN ;New individual 

  R=U(0,1) ;Draw new random number 

  CHZ(Tprev_event)=0 ;Init cum hazard 

  OBSTIME=0 ;Init obs time 

  Tprev_event=0 ;Init prev event time 

 ENDIF 

 

 ;Add a “dummy” observation each DT to 

 ;increase accuracy in A(T) 

 OBSTIME=OBSTIME+DT 

 MTIME(1)=OBSTIME 

 MTDIFF=1 

   

  

 

$DES 

 CHZ=A(T);Cum hazard at time T 

 ;Survival at time T   

 S(T)=EXP(-[CHZ-CHZ(Tprev_event)])  

 IF (R>S(T)) THEN ;if event before time T 

  Write DV=1 (event row) to sim  

  data set with time T 

  CHZ(Tprev_event)=CHZ ;Save cum hazard 

  R=U(0,1) ;Draw new random number 

 ENDIF 

$ERROR 

 IF (LIREC.EQ.NDREC) THEN ;Last ind row 

  Write DV=0 (censoring row) to sim 

  data set with time TIME 

 ENDIF 

  

 

Pseudo code for RTTE 

$PK 

 IF (NEWIND.NE.2) THEN ;New individual 

  R=U(0,1) ;Draw new random number 

  CHZ=0 ;Init cum hazard 

  OBSTIME=0 ;Init obs time 

  Tevent=-1 ;Init event time 

 ENDIF 

 

 ;Add a “dummy” observation each DT to  

 ;increase accuracy in A(T) 

 OBSTIME=OBSTIME+DT 

 MTIME(1)=OBSTIME 

 MTDIFF=1 

   

  

 

$DES 

 CHZ=A(T);Cum hazard at time T  

 ;Survival at time T 

 S(T)=EXP(-CHZ)  

 IF (R>S(T)) THEN ;if event before time T 

  Tevent=T ;Save event time 

  R=0 ;No more events 

 ENDIF 

$ERROR 

 ; If event 

 IF (Tevent.NE.-1) THEN  

  Write DV=1 (event row) to sim  

  data set with time Tevent 

 ENDIF 

 ;Last ind row and no previous event 

 IF (LIREC.EQ.NDREC.AND.Tevent.EQ.-1) THEN 

  Write DV=0 (censoring row) to sim 

  data set with time TIME 

 ENDIF 

  

 

Pseudo code for TTE 

Distribution Hazard Analytic survival time 

Exponential ℎ0 = 𝜆 𝑇𝑒𝑣𝑒𝑛𝑡 = −ln 𝑅 𝜆  

Weibull ℎ0 = 𝜆𝛼 𝜆𝑡 𝛼−1 𝑇𝑒𝑣𝑒𝑛𝑡 = − ln 𝑅
1
𝛼 𝜆  

Gompertz ℎ0 = 𝜆𝑒𝛼𝑡 
 

𝑇𝑒𝑣𝑒𝑛𝑡 =
1

𝛼
ln 1 −

𝛼 ln 𝑅

𝜆
 

Log-normal 

ℎ0 =
𝜎𝑡 2𝜋

−1
𝑒

−
1
2𝑍

2

1 − Φ 𝑍
 

𝑍 =
ln 𝑡 − 𝜇

𝜎
 

𝑇𝑒𝑣𝑒𝑛𝑡 = 𝑒𝜇+𝜎Φ
−1 1−𝑅  

Table I. 𝑡 is the time, λ is a scale parameter, 𝛼 a shape parameter, 𝜇 and 𝜎 the mean and standard deviation of a log normal 

distribution, Φ is the standard normal cumulative distribution function, Φ−1 is the inverse cumulative normal distribution 

function and 𝑅 is a uniform random number 𝑅~𝑈 0,1 . 

Figure 3. Precision of  the new MTIME method versus simulating events using a grid method for RTTE. The figure is grouped on 

the number of repeated events per subject. The events are simulated using a constant exponential hazard using a DT of 0.001 

days and shows the bias introduced (expected on average 0.25 days/event) by the grid method with twice daily observation rows. 

Figure 1. Precision in time to event relative to the analytic event time; for the MTIME method (using different DT) and the 

Grid method. The figure shows that a DT of 0.5 days gives at least the precisions seen with daily grid observations. 

The precision of the MTIME method versus the grid method when 

simulating RTTE is presented in the Figure 3.  

The data set size for the grid method was 420 times larger compared to the 

MTIME method in the RTTE example and 97 times larger in the TTE 

example. A hypothetic 3 year study with 14 000 subjects and weekly grid 

resolution yielded a data set size of ~600 MB when simulating one study 

compared to the MTIME method with a simulated data set size of ~2 MB. 

Figure 2. Pseudo code for the implementation using MTIME. 

$PROB Simulation: TTE
; Joakim Nyberg, Siv Jönsson 2014-06-01
; TTE simulation
; Dataset: 1000 subjects with TIME 0 and TIME 100
; ID TIME DV
; 1  0    0
; 1  100  0
; 2  0    0
; 2  100  0
; ....
; ....
$INPUT ID TIME DV
$DATA input_tte.csv
      IGNORE=@

$ABB COMRES=7

$SUBR ADVAN=13 TOL=9

$MODEL COMP=(HAZARD)

$PK
   ;; THE REGULAR $PK

   LAM= THETA(1)*EXP(ETA(1))
   SHP=THETA(2)

   ;; THE SIMULATION PART FOR TTE SIMULATIONS ;;

    IF (ICALL.EQ.4) THEN                 ; The event time sim $problem
          IF (NEWIND.EQ.0) THEN          ; Only for the first record
             COM(6) = 1                  ; Reset simulation ID counter
             COM(4) = 100                ; Set max time/censoring time
          ENDIF
          IF (NEWIND.EQ.1) THEN          ; For every new ind except first in dataset
             ICOUNT = COM(6) + 1         ; Update individual counter over simulations
             COM(6) = ICOUNT
          ENDIF

          IF (NEWIND.NE.2) THEN          ; For every new individual
               CALL RANDOM(2,R)



               COM(3) = -1 ; Variable for survival at event time
               COM(2) = R  ; Store the random number
               COM(1) = -1 ; Variable for the event time
               COM(7) = 0  ; Individual event counter
          ENDIF
    ENDIF

;---------MTIME for increasing precision in $DES --------

    IF (NEWIND.NE.2) THEN
     TEMP=0
    ENDIF
    TEMP=TEMP+0.1
    MTIME(1)=TEMP
    MTDIFF=1
$DES

DEL= 1E-12

;----------hazard-----------------------------------

;Weibull hazard
DADT(1)=LAM*SHP*(LAM*T+DEL)**(SHP-1)

SUR = EXP(-A(1))

IF(COM(2).GT.SUR.AND.COM(1).EQ.-1) THEN      ; If event save event time in COM(1)
   COM(1)=T
   COM(3)=SUR
ENDIF

$ERROR

"FIRST
"@CHARACTER(LEN=100)::FMT              ! Define FORMAT string for writing dataset

;; NORMAL TTE MODEL

  CHZ = A(1)
  SURX = EXP(-CHZ)           ;survival probability



   IF (COM(1).GT.COM(4)) THEN ;IF T > ENDTIME, T=ENDTIME
      ; Check survival again at endtime
      IF (COM(2).GT.SURX) THEN
         COM(1) = COM(4)
      ELSE
          COM(1) = -1 ;Integrated too far, reset event
      ENDIF
   ENDIF
   EVT = COM(1)   ; Save Event time
   RNM = COM(2)   ; Save random number, just for debugging

ENDTIME = COM(4) ; Endtime of study
TT = COM(5) ; Analytic event time

; ADD RTTE, DV TO OUTPUT, SET DV=0 IF NO EVENT OR CENSORED, DV=1 IF EVENT, RTTE = 1 IF EVENT OR CENSORED

IF (ICALL.EQ.4) THEN ; Initate DV to 0 (No event)
   DV=0
ENDIF

TMDV = 0

IF (EVID.GE.2) THEN ;Set MDV variable for output
   TMDV=1
ENDIF

ICOUNT = COM(6)+(IREP-1)*NINDR
ITER = IREP

; Define the format of the output file
"LAST
"FMT='(E13.7,8(1XE13.7))'   ! The output FORMAT

"   ! Write all events
"      IF (NEWIND.EQ.0) THEN !Open file at first record
"         OPEN (99, FILE = 'simtab.dat', POSITION='APPEND')
"          IF (IREP.EQ.1) THEN !Write header for 1st subproblem
"             WRITE (99,'(A,7(1XA))') 'ID','DV','TIME','RTTE','SURX','ICOUNT','ITER','RAND'
"          ENDIF
"       ENDIF
"       IF (EVT.NE.-1) THEN !If an EVENT



"          DV=1
"          RTTE=1
"          TMDV=0
"          ! Write SIM specific output
"          WRITE (99,FMT) ID,DV,EVT,RTTE,COM(3),ICOUNT,ITER,COM(2)
"          COM(1) = -1 !Reset Event time variable
"          COM(2) = 0  !Reset Random variable
"          COM(3) = -1 !Reset survival variable
"          COM(7) = COM(7) + 1 !Update Event counter
"       ELSE IF (LIREC.EQ.NDREC.AND.COM(7).EQ.0) THEN !Right Censoring (if no previous events)
"          DV=0
"          TMDV=0
"          RTTE=1
"          TMP=COM(4)
"          WRITE (99,FMT) ID,DV,TMP,RTTE,SURX,ICOUNT,ITER,COM(2)
"       ENDIF
"   IF (NDREC.EQ.LIREC.AND.NIREC.EQ.NINDR) THEN ! Last record for last individual
"     CLOSE(99)                                 ! Close File pointer
"   ENDIF

$THETA (0.01735)      ;1 LAMBDA
$THETA (0.8)          ;2 SHAPE ; 1 FIX for exponential distr
$OMEGA  0  FIX        ;Only to tell NONMEM that each ID has multiple rows

$SIMULATION (5988566) (39978 UNIFORM) ONLYSIM NOPREDICTION NSUB=10



$PROB Simulation: RTTE
; Joakim Nyberg, Kristin Karlsson 2014-06-01
; RTTE simulation
; Dataset: 120 subjects with TIME 0 and TIME 288
; ID TIME DV
; 1  0    0
; 1  288  0
; 2  0    0
; 2  288  0
; ....
$INPUT ID TIME DV
$DATA input_rtte.csv
      IGNORE=@
$SIML (12345) (12345 UNIFORM) ONLYSIMULATION NOPREDICTION

$THETA
(0,.00580)        ; 1 BASE

$OMEGA 0.09       ; 1 BASE

$SUBR ADVAN=13 TOL=9

$MODEL COMP=(HAZARD)

$ABB COMRES=8 ; Com variables to define RTTE variables etc.

$PK

   ;---------- Model parameters

   BASE= THETA(1)*EXP(ETA(1))

   ;---------- RTTE Simulation specifics
   IF (ICALL.EQ.4) THEN  ; The event time sim $problem
          IF (NEWIND.EQ.0) THEN          ; Only for the first record
             COM(6) = 1                  ; Reset simulation ID counter
"            ! Initialize sim output file
"            OPEN (99, FILE = 'my_data.dat', POSITION='APPEND')
"            WRITE (99,'(A,2(1XA))') 'ID','DV','TIME'

          ENDIF



          IF (NEWIND.EQ.1) THEN          ; For every new ind except first in dataset
             ICOUNT = COM(6) + 1         ; Update individual counter over simulations
             COM(6) = ICOUNT
          ENDIF

          IF (NEWIND.NE.2) THEN          ; For every new individual
               CALL RANDOM(2,R)
               COM(4) = 288; Maxtime per individual
               COM(3) = -1 ; Variable for survival at event time
               COM(2) = R  ; Store the random number
               COM(1) = -1 ; Variable for the event time
               COM(7) = 0  ; Individual event counter
               COM(8) = 0  ; Cumulative hazard
          ENDIF
    ENDIF

   ;---------MTIME for increasing $DES precision --------

    IF (ICALL.EQ.4) THEN
       IF (TIME.EQ.0) TEMP=0
       TEMP=TEMP+.1
       MTIME(1)=TEMP
       MTDIFF=1
    ENDIF

$DES

   DADT(1) = BASE

   ;---------- Calculate survival at this T
   SUR = EXP(-(A(1)-COM(8)))
   XR=0
   IF(COM(2).GT.SUR) THEN       ; If event write event to my_data.dat
       COM(1)=T                 ; Store event time
       COM(3)=SUR               ; Store survival
       COM(8)=A(1)              ; Set cumulative hazard
       COM(7)=COM(7)+1          ; Event counter
"      CALL RANDOM(2,R)
       TMP=COM(2)
       COM(2)=R                ; Store new random number



       MYDV=1
       IF (T.LE.COM(4)) THEN
"         ! Write SIM specific output
"         WRITE (99,'(E13.7,2(1XE13.7))') ID,MYDV,COM(1)
       ENDIF
   ENDIF

$ERROR

" IF (LIREC.EQ.NDREC) THEN !Right Censoring
"    DV=0
"    WRITE (99,'(E13.7,2(1XE13.7))') ID,DV,TIME
" ENDIF

"   IF (NDREC.EQ.LIREC.AND.NIREC.EQ.NINDR) THEN ! Last record for last individual
"     CLOSE(99)                                 ! Close File pointer
"   ENDIF


